
１．訂正情報                     （情報更新日：2016.04.22） 

●132 ページ・・・・・5 章 Game18「イチゴ番」 

10 行目と 60 行目を以下のように修正してください。 

ゲームは遊べますが、壁のキャラクタが←になっていました。どうも失礼致しました。 

 

【現状】 

10 R=0:POKE #700,8,15,255,128,128,240,255,8 

60 X=1:Y=1:CLS:CLP 

 

【修正】 

10 R=0:CLP:POKE #700,8,15,255,128,128,240,255,8 

60 X=1:Y=1:CLS 

 

なお、IchigoJam ver 1.2 では専用の壁文字が追加されているので、1.2 にバージョンアップ

された方は以下のプログラムを使用してください。若干雰囲気が変わります。 

 

【IchigoJam ver 1.2 用のプログラム】 

10 R=0:CLP 

20 LET[0],16392,128,17919,29728,17443,17696,4358,20486,12,6 

30 LET[10],256,16692,8258,19778,18756,18480,20736,256,12,5 

40 LET[20],0,65526,32,28710,20484,24052,36,8224,7,7 

50 LET[30],338,21632,324,21545,4,17568,8412,17408,8,5 

60 X=1:Y=1:CLS 

70 FOR I=0 TO 17:LC I,0:?CHR$(6);:LC I,9:?CHR$(6);:NEXT 

80 FOR I=1 TO 8:LC 0,I:?CHR$(6);:LC 17,I:?CHR$(6);:NEXT 

90 GOSUB 500 

100 LC X,Y:?CHR$(5); 

110 K=INKEY() 

120 G=(K=RIGHT)-(K=LEFT):A=X+G 

130 H=(K=DOWN)-(K=UP):B=Y+H 

140 V=SCR(A,B) 

150 IF X=A AND Y=B OR V=6 GOTO 110 

160 IF V=0 THEN LC X,Y:?CHR$(0);:X=A:Y=B:GOTO 100 

170 IF V<>255 GOTO 110 

180 IF SCR(A+G,B+H)<>0 GOTO 110 

190 LC D,E:?CHR$(0); 

200 D=A+G:E=B+H:LC D,E:?CHR$(255); 

210 IF D=1 AND E=1 THEN LC 6,12:?"Clear":WAIT 180:R=(R+1)&3:GOTO 60 

220 GOTO 100 

500 FOR J=0 TO 7:A=[R*10+J] 

510 FOR I=0 TO 15 

520 IF A&#8000 THEN LC I+1,J+1:?CHR$(6); 

530 A=A<<1:NEXT:NEXT 

560 D=[R*10+8]:E=[R*10+9]:LC D,E:?CHR$(255); 

570 RETURN 

 

2．著者関連のリンク 

・本書の著者の一人、古籏一浩氏が開設している IchigoJam に関するあれやこれやのサイトは

コチラ。 

 http://www.openspc2.org/reibun/IchigoJam/  

 

 

［先頭ページに戻る］ 

http://www.openspc2.org/reibun/IchigoJam/book/index.html

